

Descobrimos o Dia da Semana

Mário Leite

...

Uma das curiosidades mais interessantes, e que às vezes torna-se muito importante para as pessoas, é saber em que dia da semana “caiu” ou “vai cair” determinada data. Por exemplo, saber em que dia da semana “caiu” 06/10/1990; se o calendário estiver disponível ou paciência para procurar no Windows® ou no Google, veremos que foi num Sábado. Mas... e a data de 06/06/1944 (dia D - desembarque dos aliados nas praias da Normandia)? Ou em que dia da semana o homem pisou pela primeira vez na Lua? Ou mesmo em que dia da semana você nasceu? É claro que pesquisando na Internet é possível descobrir essas datas e o respectivo dia da semana se o buscador mostrar; mas, o mais interessante e mais estimulante para um programador é CALCULAR o dia da semana pela fórmula abaixo, e aproveitando para exercitar com um pequeno programa. Mas, ocorrem duas exceções em relação aos meses de Janeiro e Fevereiro. Nestes casos, se o mês for Janeiro **m** deverá ter o valor **13** e não o valor da data considerada; e se o mês for Fevereiro **m** tem que ser **14**. Nestes dois casos (Janeiro e Fevereiro) o ano deverá ser subtraído de 1. Por exemplo, **30/01/1992**; então tem que fazer: **m=13** e **a=1991**; o dia continua **d=30**; e para esta data encontraremos o dia da semana como sendo uma *Quinta-Feira*.

$$F = d + 2*m + \text{int}((3(m+1))/5) + a + \text{int}(a/4) - \text{int}(a/100) + \text{int}(a/400) + 2$$

- d : dia
- m : mês
- a : ano
- int : pega apenas a parte inteira do resultado nas operações de divisão.

Depois que calcular **F**, basta calcular o dia como sendo *resto* da divisão de **F** por **7**.

Dia = (F Resto 7)

Em função de Dia é definido o dia da semana:

Dia = 0 ==> Sábado
Dia = 1 ==> Domingo
Dia = 2 ==> Segunda-feira
Dia = 3 ==> Terça-feira
Dia = 4 ==> Quarta-feira
Dia = 5 ==> Quinta-feira
Dia = 6 ==> Sexta-feira

Observe o exemplo abaixo:

Data: 20/07/1969

a = 1969

m = 07

d = 20

$F = 20 + 2*7 + \text{int}((3*(7+1))/5) + 1969 + \text{int}(1969/4) - \text{int}(1969/100) + \text{int}(1969/400) + 2$

F = 2486

(2486 **Resto 7**)=1 ==>**Domingo** (dia em que o homem pisou na Lua pela primeira vez)

O programa “DiaDaSemana”, em pseudocódigo, a seguir, mostra uma solução simples em pseudocódigo para determinar o dia da semana de uma determinada data.


```

Programa DiaDaSemana
//Dá o dia da semana de uma data
//Em Pseudocódigo
//Autor: Mário Leite
//-----
Declare d, m, a, F, Dia: inteiro
      DiaS: caractere
INÍCIO
  Escreva("Digite o dia: ")
  Leia(d)
  Escreva("Digite o mês: ")
  Leia(m)
  Escreva("Digite o ano: ")
  Leia(a)
  Se(m=01) Então
    m ← 13
    a ← a - 1
  FimSe
  Se(m=02) Então
    m ← 14
    a ← a - 1
  FimSe
  F ← d + 2*m + Int((3*(m+1))/5) + a +Int(a/4) -
  Int(a/100) + Int(a/400)+ 2
  Dia ← (F Resto 7)
  Selecione Dia
    Caso 0
      DiaS ← "Sábado"
    Caso 1
      DiaS ← "Domingo"
    Caso 2
      DiaS ← "Segunda-feira"
    Caso 3
      DiaS ← "Terça-feira"
    Caso 4
      DiaS ← "Quarta-feira"
    Caso 5
      DiaS ← "Quinta-feira"
    Caso 6
      DiaS ← "Sexta-feira"
  FimSelecione
  EscrevaLn("")
  EscrevaLn("O dia da semana é: ", DiaS)
FIM

```

A figura 1 mostra a saída do programa testado em Visualg.

Nota: No programa “DiaDaSemana” acima não foram validados de **d**, **m** e **a** (dia, mês e ano). No caso de um programa mais rigoroso estes dados têm que ser validados: (**0<d≤31**, **0<m≤12** e **a** tem que pertencer ao calendário Gregoriano - partir de 1583)



```
Digite o dia: 20
Digite o mês: 07
Digite o ano: 1969

Dia da semana: Domingo

*** Fim da execução.
*** Feche esta janela para retornar ao Visualg.
```

Figura 1 - Saída do programa “DiaDaSemana” testado e codificado em Visualg