

Função de Euler - Coprimos

Mário Leite

...

De acordo com o matemático americano nascido na Rússia, Edward Frenkel, função de Euler, também conhecida como função Totiente ou função $\phi()$ - Função Fi - de um número natural N é $\phi(N)=y$ onde y é um número natural que representa a quantidade de números inteiros x_i [$i=1,y-1$], tal que $MDC(x_i,y)=1$. Baseado no “Teorema de Euler”, isto quer dizer que y dá a quantidade de coprimos com N no intervalo $1 \leq i < y$. Por exemplo, para $N=9$ $\phi(9)=6$ porque de 1 até (9-1) existem seis coprimos com 9: **1 2 4 5 7, 8**; isto significa que:

$$MDC(1,9) = 1$$

$$MDC(2,9) = 1$$

$$MDC(4,9) = 1$$

$$MDC(5,9) = 1$$

$$MDC(7,9) = 1$$

$$MDC(8,9) = 1$$

De acordo com a apresentação mostrada por Pimentel^[1] (2010) existe um teorema sobre a Função de Euler que diz:

“Se a e b são inteiros positivos tais que $MDC(a,b) = 1$, então $\phi(ab) = \phi(a)\phi(b)$ ”.*

O conceito de coprimo é fundamental na criptografia do método RSA de chave dupla. selecionando um número e que seja coprimo com $\phi(n)$ para criar a chave pública (e,n) que irá codificar mensagens. Este recurso também é utilizado para codificar senhas de cartões de crédito; por isto este assunto é muito importante.


O programa “**ProgFuncaoEuler**”, codificado em Visualg (que pode ser convertido para qualquer outra linguagem) mostra como calcular e exibir a função de Euler de um número inteiro positivo, além de exibir também os coprimos com esse número. As **figuras 1, 2, 3** mostram os resultados desta função para três diferentes números: **6, 11 e 66**, respectivamente. Note que, além disto, na **figura 3**, que $\phi(66) = \phi(6)*\phi(11) = 2*10 = 20$, comprovando o “Teorema de Euler”.

¹ PIMENTEL, Elaine. “Algebra A - Função de Euler”, Departamento de Matemática, UFMG, 2º Semestre, 2010. Material publicado em PDF disponível no link www.mat.ufmg.br/~elaine/AlgebraA/aula06_2010.pdf (acesso em 21/05/15 - 17:49)

```

Algoritmo "ProgFuncaoEuler"
//Determina a função de Euler  $\phi(n)$  e os coprimos com d menores que n.
//Em Visualg
//Autor: Mário Leite
//-----
Var j, n, Fi, Cont, MDC: inteiro
VetPrimos: vetor[1..1000] de inteiro
F: caractere
//-----
funcao FuncEuler(d:inteiro): inteiro
{Verifica se o número é coprimo com n e conta}
var j, Aux: inteiro
inicio
Para j De 1 Ate (d-1) Faca
MDC <- FuncMDC(d,j) //chama função para calcular o MDC(d,j)
Se (MDC=1) Entao
Cont <- Cont + 1
VetPrimos[Cont] <- j
FimSe
FimPara
Retorne Cont //retorna quantidade de coprimos com d, menores que d
Fimfuncao //fim da função "FuncEuler"
//-----
Funcao FuncMDC(N1,N2:inteiro): inteiro
{Calcula o MDC(d,j) utilizando o algoritmo de Euclides}
var Aux: inteiro
inicio
Enquanto (N2<>0) Faca
Aux <- N1
N1 <- N2
N2 <- (Aux Mod N2)
FimEnquanto
MDC <- N1
Retorne MDC //retorna o MDC(d,j)
fimFuncao //fim da função "FuncMDC"
//=====
//Programa principal
Inicio
LimpaTela
Cont <- 0
Escreva("Entre com um número inteiro e positivo: ")
Leia(n)
Enquanto (n<0) Faca
Escreva("Entre com um número inteiro e positivo: ")
Leia(d)
FimEnquanto
Escreval("")
{Trata a exceção para d=1}
Se(n=1) Entao
Fi <- 1
F <- "1"
Escreval("Números coprimos com ", F, ": ")
Senao
Fi <- FuncEuler(n) //obtem a função de Euler:  $\phi(n)$ 
F <- NumpCarac(n)
Escreval("Números coprimos com ", F, " (menores que", n, "):")
FimSe
Para j De 1 Ate Cont Faca
Escreval(VetPrimos[j])
FimPara
Escreval("")
Escreval("Função de Euler Fi(", F, "):", Fi)
FimAlgoritmo //fim do programa "ProgFuncaoEuler"

```



```
Console simulando o modo texto do MS-DOS


Entre com um número inteiro e positivo: 6

Números coprimos com 6 (menores que 6):
1
5

Função de Euler Fi(6): 2

>>> Fim da execução do programa !
```

Figura 1 - Função de Euler para n=6



```
Console simulando o modo texto do MS-DOS


Entre com um número inteiro e positivo: 11

Números coprimos com 11 (menores que 11):
1
2
3
4
5
6
7
8
9
10

Função de Euler Fi(11): 10

>>> Fim da execução do programa !
```

Figura 2 - Função de Euler para n=11



```
Console simulando o modo texto do MS-DOS

Entre com um número inteiro e positivo: 66

Números coprimos com 66 (menores que 66):
1
5
7
13
17
19
23
25
29
31
35
37
41
43
47
49
53
59
61
65

Função de Euler Fi(66): 20

>>> Fim da execução do programa !
```

Figura 3 - Função de Euler para n=66