


## Produto Misto

Mário Leite

...

O Produto Misto, ao contrário dos produtos escalar e vetorial, não envolve apenas dois vetores, mas, sim três. Então, vamos considerar três vetores **U**, **V**, **W** como mostrado no esquema da **figura 1**. O “Produto Misto” (Pm) é um número real dado por  $Pm = U_o(V \times W)$ ; deste modo, pode ser calculado pelo produto escalar do vetor **U** pelo produto vetorial de **V** por **W**.



**Figura 1 - Esquema espacial de três vetores no “Produto Misto”**

Considerando o esquema da **figura 1** temos a seguinte expressão:

$$Pm = [U, V, W] = U * |V \times W| * \cos(\theta)$$

Onde  $|V \times W|$  é o módulo (valor absoluto) do produto vetorial de **V** por **W**.

Semelhantemente ao dispositivo matricial usado para calcular o produto vetorial, o “Produto Misto” também pode ser indicado pelo mesmo dispositivo, com a diferença que a primeira linha da matriz não é constituída pelos vetores unitários e sim pelos elementos do vetor **U** como mostrado abaixo.


	$u_1$	$u_2$	$u_3$
$U_o(V \times W)$	$=$	$v_1 \ v_2 \ v_3$	
		$w_1 \ w_2 \ w_3$	

Analiticamente, para o cálculo do Pm deve ser colocado assim:

$u_1$	$u_2$	$u_3$	$u_1$	$u_2$
$v_1$	$v_2$	$v_3$	$v_1$	$v_2$
$w_1$	$w_2$	$w_3$	$w_1$	$w_2$

$$Pm = (u_1 * v_2 * w_3 + u_2 * v_3 * w_1 + u_3 * v_1 * w_2) - (u_3 * v_2 * w_1 + u_1 * v_3 * w_2 + u_2 * v_1 * w_3)$$

Uma aplicação prática do “Produto Misto” é no cálculo do volume de um paralelepípedo. Este volume é numericamente igual ao produto misto dos vetores que determinam suas três arestas. Assim, se o paralelepípedo tem arestas definidas pelos vetores **A**, **B** e **C**, então seu volume pode ser dado por:  $A_o(B \times C)$ . O programa “**ProdutoMisto**” (codificado em Visualg) mostra uma solução para calcular o produto misto entre os vetores **U**, **V** e **W** cujas respectivas coordenadas são três elementos inteiros entrados pelo teclado. A **figura 1** mostra um exemplo de saída deste programa quando se calcula o “Produto Misto” destes vetores.



```
Leitura dos elementos do vetor U
Digite o elemento [ 1] do vetor U: 2
Digite o elemento [ 2] do vetor U: 3
Digite o elemento [ 3] do vetor U: 5

Leitura dos elementos do vetor V
Digite o elemento [ 1] do vetor V: -1
Digite o elemento [ 2] do vetor V: 3
Digite o elemento [ 3] do vetor V: 3

Leitura dos elementos do vetor W
Digite o elemento [ 1] do vetor W: 4
Digite o elemento [ 2] do vetor W: -3
Digite o elemento [ 3] do vetor W: 2

Produto Misto Uo(VxW) = 27

>>> Fim da execução do programa !
```

Figura 1 - Um exemplo de saída do programa “ProdutoMisto”

```
Algoritmo "ProdutoMisto"
//Faz o "Produto Misto" de dois vetores no R³.
//Em Visualg
//Autor: Mário Leite
//-----
Var U, V, W, Pv: vetor[1..3] de inteiro
    i, Pm: inteiro
Inicio
    {Leitura do primeiro vetor}
    Escreval("Leitura dos elementos do vetor U")
    Para i De 1 Ate 3 Faca
        Escreva("Digite o elemento [",i,"] do vetor U: ")
        Leia(U[i])
    FimPara
    Escreval("")
    {Leitura do segundo vetor}
    Escreval("Leitura dos elementos do vetor V")
    Para i De 1 Ate 3 Faca
        Escreva("Digite o elemento [",i,"] do vetor V: ")
        Leia(V[i])
    FimPara
    Escreval("")
    {Leitura do terceiro vetor}
    Escreval("Leitura dos elementos do vetor W")
    Para i De 1 Ate 3 Faca
        Escreva("Digite o elemento [",i,"] do vetor W: ")
        Leia(W[i])
    FimPara
    Escreval("")
    Escreval("")
    {Calcula e exibe o produto misto Uo(VxW)}
    Escreva("Produto Misto Uo(VxW) =")
    Pm <- U[1]*V[2]*W[3] + U[2]*V[3]*W[1] + U[3]*V[1]*W[2]
    Pm <- Pm-(U[3]*V[2]*W[1]+U[1]*V[3]*W[2]+ U[2]*V[1]*W[3])
    Pm <- Abs(Pm) //Pm não pode ser negativo
    Escreval(Pm)
    Escreval("")
FimAlgoritmo
```