

## Validando CNPJ

Mário Leite

Em todos os cursos de programação que já ministrei, sempre é feita a seguinte pergunta:

“Como podemos validar o CNPJ (antigo CGC) uma empresa?” Esta é uma questão que deve sempre merecer atenção especial por parte dos programadores profissionais, pois a validação desse dado do cliente é fundamental no processo de cadastramento. E mesmo que o Google possa responder a esta pergunta, apresentando as famosas “calculadoras de checagem”, o programador deve saber como fazer isto de maneira simples e inteligível, já que os códigos mostrados na Internet são confusos e não explicam como fazer. Uma das maneiras de verificar se o CNPJ está correto é verificar os seus dois **dígitos verificadores** (aqueles dois números logo após o traço); são eles que confirmam o CNPJ. Por exemplo, vamos supor (hipoteticamente) que o CNPJ de uma determinada empresa seja 12.345.678/0009-57. Assim, para checar se tal CNPJ está correto basta verificar se o primeiro DV (Dígito Verificador) é **5** e se o segundo é **7**; se qualquer um dos dois não for validado algo está errado com esse CNPJ. E como podemos fazer isto? Normalmente, os dígitos verificadores são calculados através de um sistema de algoritmo denominado “**Módulo 11**”, o que vamos fazer na prática, com um esquema de algoritmo bem didático, considerando o exemplo de CNPJ dado: 12.345.678/0009-57. Primeiramente colocamos este valor só com os números e sem os dois dígitos verificadores da seguinte maneira: **123456780009**. Numa segunda etapa multiplicamos cada algarismo deste número por **2**, por **3**, por **4**, etc (de trás para frente) até **9**; uma vez chegando ao **9** recomeçamos novamente a multiplicação por **2**, por **3** por **4** e por **5**. Depois destas multiplicações somamos as duas filas de números (o CNPJ + os produtos). Em terceiro lugar dividimos essa soma por **11** e consideramos o resto dessa divisão. Neste caso o primeiro dígito verificador será **11 - resto**. Mas, devemos considerar uma exceção a esta regra: se o resultado dessa subtração for **10** ou **11** o dígito verificador será **0** (por definição). Assim, obtemos o primeiro dígito; e para calcular o segundo dígito basta considerar, agora, o CNPJ com 13 dígitos (incluindo o primeiro dígito calculado e começando a multiplicação também por **2**). A **figura 1** ilustra o processo do “**Módulo 11**”.






Figura 1 - Algoritmo “Módulo 11” para calcular Dígito Verificador


Então, como visto na **figura 1** os dígitos verificadores do nosso *CNPJ* hipotético são **4** e **2** e não **5** e **7** como havia sido considerado; desse modo, o *CNPJ* correto seria **12.345.678/0009-42**. E como podemos automatizar isso? É muito fácil, basta escrever uma rotina que traduza o esquema de algoritmo dessa figura, traduzido no programa “**VerificaCNPJ**”, implementado em VB6. A **figura 2.1** mostra o *design* da interface do programa, antes de ser executado. As **figuras 2.2, 2.3** e **2.4** mostram o programa em execução (rodando). Na **figura 2.2** a aplicação está rodando (depois de pressionar **F5**), mas sem qualquer disparo de ação na interface. A **figura 2.3** mostra ações na interface: o usuário digitou um valor de *CNPJ* na primeira caixa de texto e clicou no botão [**Executar**]; o resultado foi o aviso de “**CNPJ INÁLIDO!**”. Finalmente, na **figura 2.4** a resposta do sistema dá como válido o *CNPJ* digitado, conforme havia sido previsto anteriormente. Além das chamadas *procedures-evento*, que implementam as ações do usuário na interface em execução, a aplicação é baseada em duas funções: “**Modulo11()**” que retorna os dígitos verificadores baseado no esquema da **figura 1**, e a função **VerifCNPJ()** que recebe o número do *CNPJ* (completo - com os dois DV’s) e retorna **True** se esses dois DV’s estiverem corretos ou **False** em caso contrário. Observe que esta função faz duas chamadas à função **Modulo11()**; na primeira verifica o primeiro dígito e na segunda chamada verifica o outro dígito. Desse modo seu retorno só será **True** se ambos os DV’s estiverem ambos, corretos.




**Figura 2.1 - Aplicação em tempo de projeto**



**Figura 2.2 - Aplicação rodando: sem qualquer ação na interface**



**Figura 2.3 - Aplicação rodando: click no botão [Executar] - CNPJ inválido**



**Figura 2.4 - Aplicação rodando: click no botão [Executar] - CNPJ válido**

```
'Verifica a validade de um CNPJ
'Em VB6
'Autor: Mário Leite
'Data: 05/02/2023
'-----

Public Function VerifCNPJ(num) As Boolean
    'Função para validar o CNPJ
    If (Len(num) <> 14) Then
        Beep 'emite um apito de alerta
        MsgBox ("Tamanho do CNPJ é diferente de 14", vbCritical + vbOK, "Erro")
        Exit Function
    End If
    'Valida o primeiro dígito
    If (Modulo11(Left(num, 12)) <> Mid(num, 13, 1)) Then 'chama função
        VerifCNPJ = False
        Exit Function
    End If
    'Valida o segundo dígito e dá o retorno da função
    If (Modulo11(Left(num, 13)) <> Mid(num, 14, 1)) Then 'chama função
        VerifCNPJ = False
    Else
        VerifCNPJ = True 'retorno da função
    End If
End Function
'-----

Public Function Modulo11(NúmeroCNPJ) As String
    'Função para validar os dígitos verificadores
    Dim i As Integer, Produto As Integer, Multiplicador As Integer, Dígito As Integer
    Multiplicador = 2
    For i = Len(NúmeroCNPJ) To 1 Step -1
        Produto = Produto + Val(Mid(NúmeroCNPJ, i, 1)) * Multiplicador
        Multiplicador = IIf(Multiplicador = 9, 2, Multiplicador + 1)
    Next i
    'Considera exceção
    Dígito = 11 - Int(Produto Mod 11)
    Dígito = IIf(Dígito = 10 Or Dígito = 11, 0, Dígito)
    Modulo11 = Trim(Str(Dígito)) 'Retorno da função
End Function
'-----
```

```

Private Sub btnExecutar_Click(Index As Integer)
    Dim strCNPJ As String, Pos1 As String, Pos2 As String
    Dim Pos3 As String, Pos4 As String, Pos5 As String
    Dim CNPJFormatado As String
    Dim ret As Boolean

    txtValidacao.FontBold = True

    'Valida a entrada
    strCNPJ = Trim(txtCNPJ.Text)
    Pos1 = Left(strCNPJ, 2)
    Pos2 = Mid(strCNPJ, 3, 3)
    Pos3 = Mid(strCNPJ, 6, 3)
    Pos4 = Mid(strCNPJ, 9, 4)
    Pos5 = Right(strCNPJ, 2)
    CNPJFormatado = (Pos1 + "." + Pos2 + "." + Pos3 + "/" + Pos4 + "-" + Pos5)

    txtFormatado.Text = CNPJFormatado
    'Valida o CNPJ
    If (IsNumeric(strCNPJ)) Then 'CNPJ só números?
        ret = VerifCNPJ(strCNPJ) 'chama função
        If (ret) Then
            txtValidacao.ForeColor = vbBlue
            txtFormatado.ForeColor = vbBlue
            txtValidacao.Text = "CNPJ válido."
        Else
            txtValidacao.ForeColor = vbRed
            txtFormatado.ForeColor = vbRed
            txtValidacao.Text = "CNPJ INVÁLIDO!"
        End If
    Else 'valor digitado não é numérico
        MsgBox ("CNPJ tem que ter só números!!!")
        txtCNPJ.Text = ""
        txtCNPJ.SetFocus
    End If
End Sub
'-----

Private Sub btnFechar_Click(Index As Integer)
    End 'encerra a aplicação
End Sub
'-----

Private Sub btnLimpar_Click(Index As Integer)
    'Limpa as caixas de texto para nova verificação
    txtCNPJ.Text = ""
    txtValidacao.Text = ""
    txtCNPJ.SetFocus
End Sub
'-----

Private Sub Form_Activate()
    txtCNPJ.SetFocus 'focaliza a primeira caixa de texto na ativação do formulário
End Sub

```