

Calculando Determinante de Matrizes

Mário Leite

...

Como já mencionei em postagens passadas, o determinante de uma matriz é um dos elementos mais importante da Matemática. Definido como sendo “*um número real associado à uma matriz quadrada*”, o determinante está presente em diversas situações de cálculos complexos para em que é necessário para definir elementos importantes na nossa vida diária, e até mesmo na Matemática Avançada; por exemplo para verificar se determinada matriz é ou não invertível, para verificar se três pontos estão alinhados num plano cartesiano, para calcular área de um triângulo, na resolução de sistemas de equações lineares, nos cálculos de autovalores e autovetores, etc.

Na literatura impressa e nas pesquisas através de buscadores na Internet podemos encontrar muitos exemplos sobre determinante como calcular. Entretanto, na maioria das vezes estes exemplos se resumem a determinantes de matrizes de ordem três, no máximo. Para matrizes de ordem superior a três existem as chamadas “calculadoras *on line*” que solicitam os elementos da matriz e calculam o determinante; bem interessante e funcional. Entretanto, para quem quer aprender COMO CALULAR a coisa fica complicada pois isto permanece um “mistério”. Pensando nesta situação, resolvi criar um programa para calcular o determinante de matrizes de qualquer ordem. O programa “**ProgDeterminante**” (em Visualg) calcula e mostra o determinante de matrizes de qualquer ordem, a partir de ordem 2 e de maneira bem compreensível e bem comentado, com três módulos, além do programa principal:

FunCalculaDet2x2 Função que retorna o determinante de matrizes 2x2.

FunCalculaDet3x3 Função que retorna o determinante de matrizes 3x3.

FunCalculaDetmxm Função que retorna o determinante de matrizes mxm ($m > 3$).

E para testar seus conhecimentos básicos sobre determinantes, explique:

“Por que o valor do determinante da figura 1 deu valor 2, enquanto que o da figura 2 deu 0; este valor 0 foi apenas uma coincidência ou teria que ser este valor mesmo!?”

Algoritmo "ProgDeterminante"

```
//Calcula o determinante de uma matriz mxm de inteiros.
//Em Visualg
//Autor: Mário Leite
//E-mail : marleite@gmail.com
//-----
//Elementos globais
Const MAXLIN=10 /limita as dimensões da matriz (pode ser maior).
Var MatOrig, MatA, MatX, Mat3x3: vetor[1..MAXLIN,1..MAXLIN] de real
    p, q, mLin, LinMat: inteiro
    SomaPos, SomaNeg, DetA: real
//-----
Funcao FunDet2x2: real
//Calcula o determinante de ordem 2 na forma mais simples
    var Det2: real
inicio
    SomaPos <- MatA[1,1]*MatA[2,2]
    SomaNeg <- MatA[2,1]*MatA[1,2]
    Det2 <- SomaPos - SomaNeg
    Retorne Det2
FimFuncao //fim da função "FunDet2x2"
//-----
Funcao FunDet3x3: real
//Calcula o determinante de ordem 3 pela "Regra de Sarrus"
    var Det3: real
inicio
    SomaPos <- MatA[1,1]*MatA[2,2]*MatA[3,3] + MatA[1,2]*MatA[2,3]*MatA[3,1]
    SomaPos <- SomaPos + MatA[1,3]*MatA[2,1]*MatA[3,2]
    SomaNeg <- MatA[1,3]*MatA[2,2]*MatA[3,1] + MatA[1,1]*MatA[2,3]*MatA[3,2]
    SomaNeg <- SomaNeg + MatA[1,2]*MatA[2,1]*MatA[3,3]
    Det3 <- SomaPos - SomaNeg
    Retorne Det3
FimFuncao //fim da função "FunDet3x3"
//-----
Funcao FunDetmxm: real
//Calcula o determinante de ordem superior a 3
    var MatAx: vetor[0..MAXLIN, 0..MAXLIN] de real
        i, j, k, m, Trocas: inteiro
        Detmxm, Aux, Fator: real
inicio
    {Faz uma cópia da matriz lida para o processamento}
    Para i De 0 Ate (LinMat-1) Faca
        Para j De 0 Ate (LinMat-1) Faca
            MatAx[i,j] <- MatA[i+1,j+1]
        FimPara
    FimPara
    Trocas <- 0 //número de permutas de linhas da matriz
    {Transforma a matriz num triangulo para utilizar a "Regra de Chió"}
    m <- LinMat
    Para i De 0 Ate (m-2) Faca
        Se (MatAx[i,i]=0) Entao //detectou o elemento [1,1] como 0
            Para k De i Ate (m-1) Faca //procura um elemento [1,1] diferente de 0
                Se (MatAx[k,i]<>0) Entao
```

```

        Para j De 0 Ate (m-1) Faca //permuta linhas
            Aux <- MatAx[i,j]
            MatAx[i,j] <- MatAx[k,j]
            MatAx[k,j] <- Aux
        FimPara
        k <- m
    FimSe
FimPara
Trocas <- Trocas + 1 //acumula o número de permutas de linhas
FimSe
Se (MatAx[i,i] <> 0) Entao
    {Converte o elemento [1,1] para 1}
    Para k De (i+1) Ate (m-1) Faca
        Fator <- -1*MatAx[k,i]/MatAx[i,i]
        Para j De i Ate (m-1) Faca
            MatAx[k,j] <- MatAx[k,j] + (Fator*MatAx[i,j])
        FimPara
    FimPara
FimSe
FimPara
FimSe
FimPara
Detmxm <- 1.0
{Calcula o determinante}
Para i De 0 Ate (m-1) Faca
    Detmxm <- Detmxm*MatAx[i,i]
FimPara
Escreval("")
Escreval("")
Se (Trocas Mod 2 <> 0) Entao //permuta ímpar de linhas
    Detmxm <- -Detmxm
FimSe
Retorne Detmxm
FimFuncao //fim da função "FunDetmxm"

```

//=====

//Programa principal

Inicio

```

Repita //leitura e validação da matriz
    Escreva("Digite o número de linhas da matriz: ")
    Leia(LinMat)
    LinMat <- Int(LinMat)
Ate((LinMat>=2) e (LinMat<=MAXLIN))
Escreval("")
Escreval("")
Para p De 1 Ate LinMat Faca
    Para q De 1 Ate LinMat Faca
        Escreva("Entre com o elemento [",p,q,"] da matriz: ")
        Leia(MatOrig[p,q]) //matriz original lida preservada
        mLin <- LinMat //preserva o número de linhas da matriz original
        MatA[p,q] <- MatOrig[p,q] //faz cópia da matriz para usar nos cálculos
    FimPara
    Escreval("")
FimPara

```

```

{Seleciona o tipo de função a ser chamada para calcular o determinante}
Escolha LinMat
  Caso 2
    DetA <- FunDet2x2    //para matrizes de ordem 2
  Caso 3
    DetA <- FunDet3x3    //para matrizes de ordem 3
  OutroCaso
    DetA <- FunDetmxm    //para matrizes de ordem superior a 3
FimEscolha
LimpaTela
{Mostra a matriz na forma tabular}
Escreval(" Matriz original lida:")
Para p De 1 Ate mLin Faca
  Para q De 1 Ate mLin Faca
    Escreva(MatOrig[p,q], " ")
  FimPara
  Escreval("")
FimPara
Escreval("")
Escreval("")
Escreval("Determinante da matriz:", DetA)
Escreval("")
FimAlgoritmo //fim do programa "ProgDeterminante"

```

```

Digite o número de linhas da matriz: 5

Entre com o elemento [ 1 1] da matriz: 1
Entre com o elemento [ 1 2] da matriz: 2
Entre com o elemento [ 1 3] da matriz: 3
Entre com o elemento [ 1 4] da matriz: 4
Entre com o elemento [ 1 5] da matriz: 0

Entre com o elemento [ 2 1] da matriz: 2
Entre com o elemento [ 2 2] da matriz: 4
Entre com o elemento [ 2 3] da matriz: 6
Entre com o elemento [ 2 4] da matriz: 8
Entre com o elemento [ 2 5] da matriz: 1

Entre com o elemento [ 3 1] da matriz: 1
Entre com o elemento [ 3 2] da matriz: 3
Entre com o elemento [ 3 3] da matriz: 5
Entre com o elemento [ 3 4] da matriz: 7
Entre com o elemento [ 3 5] da matriz: 9

Entre com o elemento [ 4 1] da matriz: 4
Entre com o elemento [ 4 2] da matriz: 4
Entre com o elemento [ 4 3] da matriz: 6
Entre com o elemento [ 4 4] da matriz: 8
Entre com o elemento [ 4 5] da matriz: 0


Entre com o elemento [ 5 1] da matriz: 3
Entre com o elemento [ 5 2] da matriz: 5
Entre com o elemento [ 5 3] da matriz: 7
Entre com o elemento [ 5 4] da matriz: 8
Entre com o elemento [ 5 5] da matriz: 1

Matriz original lida:
1 2 3 4 0
2 4 6 8 1
1 3 5 7 9
4 4 6 8 0
3 5 7 8 1

Determinante da matriz: 2

*** Fim da execução.
*** Feche esta janela para retornar ao Visualg.
```

Figura 1 - Determinante com valor 2



```
Digite o número de linhas da matriz: 5

Entre com o elemento [ 1 1] da matriz: 1
Entre com o elemento [ 1 2] da matriz: 2
Entre com o elemento [ 1 3] da matriz: 3
Entre com o elemento [ 1 4] da matriz: 4
Entre com o elemento [ 1 5] da matriz: 0

Entre com o elemento [ 2 1] da matriz: 2
Entre com o elemento [ 2 2] da matriz: 4
Entre com o elemento [ 2 3] da matriz: 6
Entre com o elemento [ 2 4] da matriz: 8
Entre com o elemento [ 2 5] da matriz: 0

Entre com o elemento [ 3 1] da matriz: 1
Entre com o elemento [ 3 2] da matriz: 3
Entre com o elemento [ 3 3] da matriz: 5
Entre com o elemento [ 3 4] da matriz: 7
Entre com o elemento [ 3 5] da matriz: 9

Entre com o elemento [ 4 1] da matriz: 4
Entre com o elemento [ 4 2] da matriz: 4
Entre com o elemento [ 4 3] da matriz: 6
Entre com o elemento [ 4 4] da matriz: 8
Entre com o elemento [ 4 5] da matriz: 0

Entre com o elemento [ 5 1] da matriz: 3
Entre com o elemento [ 5 2] da matriz: 5
Entre com o elemento [ 5 3] da matriz: 7
Entre com o elemento [ 5 4] da matriz: 8
Entre com o elemento [ 5 5] da matriz: 1

Matriz original lida:
1  2  3  4  0
2  4  6  8  0
1  3  5  7  9
4  4  6  8  0
3  5  7  8  1

Determinante da matriz: 0

*** Fim da execução.
*** Feche esta janela para retornar ao Visualg.
```

Figura 2 - Determinante com valor 0