

Anos Palíndromos

Mário Leite

...

Palíndromos são sequência de caracteres que podem ser lidas tanto da esquerda para a direita quanto da direita para a esquerda. Assim, por exemplo, as palavras “ana”, “radar” e “omissíssimo” são palíndromos. Existem até frases consideradas palíndromos, como a famosa “Socorram-me! Subi no ônibus em Marrocos”. Mas, e com relação a números; também existem números palíndromos? Sim. Existem números palíndromos, tal como as palavras, mas são chamados de “números capicuas”. O programa “**MostraAnosPalindromos**” codificado em Visualg e mostrado abaixo mostra os anos palíndromos, transformados em “números capicuas” dentro de uma faixa de anos determinada pelo usuário. Note que devido ser parecido com pseudocódigo, este programa poderá ser convertido em qualquer linguagem. A **figura 1** mostra a saída do programa, listando os anos palíndromos desde 1000 até 3000.

Algoritmo "MostraAnosPalindromos"

//Mostra os anos palíndromos numa faixa de tempo determinada pelo usuário.

//Autor: Mário Leite

//-----

Var i, j, Cont, Tam, Lim1, Lim2: inteiro

Car, AnoStrD, AnoStrI: caractere

Início

Repita //valida as entradas dos limites

Escreva(" Entre com o limite inferior da faixa [min 10]: ")

Leia(Lim1)

Lim1 <- **Int**(**Abs**(Lim1)) //garante limite inferior inteiro e positivo

Escreva(" Entre com o limite superior da faixa: ")

Leia(Lim2)

Lim2 <- **Int**(**Abs**(Lim2)) //garante limite superior inteiro e positivo

Escreval(" ")

Ate((Lim2>Lim1) e (Lim1>=10))

//LimpaTela

Escreval(" ") //salta linha

Escreval(" Anos palíndromos desde",Lim1, " até", Lim2,":")

Cont <- 0

Para i **De** Lim1 **Ate** Lim2 **Faca**

AnoStrD <- **NumpCarac**(i) //converte número para string

Tam <- **Compr**(AnoStrD) //pega o tamanho da string

AnoStrI <- ""

Para j **De** 1 **Ate** Tam **Faca**

Car <- **Copia**(AnoStrD, (Tam-j+1),1) //copia dígito num string

AnoStrI <- AnoStrI + Car

FimPara

{Verifica se o ano é um palíndromo}

Se(AnoStrD=AnoStrI) **Entao**

Escreval(" ", AnoStrD) //imprime o ano palíndromo

Cont <- Cont + 1

FimSe


FimPara

Escreval(" ")

Escreval(" Existem", Cont, " anos palíndromos desde",Lim1, " até", Lim2,":")

Escreval(" ")

FimAlgoritmo



```
C:\> Console simulando o modo texto do MS-DOS

Entre com o limite inferior da faixa [min 10]: 1000
Entre com o limite superior da faixa: 3000

Anos palíndromos desde 1000 até 3000:
1001
1111
1221
1331
1441
1551
1661
1771
1881
1991
2002
2112
2222
2332
2442
2552
2662
2772
2882
2992

Existem 20 anos palíndromos desde 1000 até 3000.

>>> Fim da execução do programa !
```

Figura 1 - Lista dos anos palíndromos desde 1000 até 3000