

O Poderoso SciLab - Parte I

Mário Leite

...

O SciLab (abreviação de **Scientific Laboratory**)) é um *software* de código aberto lançado em 1990; criado inicialmente pelo INRIA (Instituto Nacional de Pesquisa em Ciência da Computação e Automação, França). Seu objetivo principal era oferecer uma alternativa gratuita ao MATLAB, para cálculos numéricos e computação científica, voltado para engenheiros, cientistas, professores e estudantes. Em 2003, sua manutenção passou para a “Scilab Enterprises”, empresa dedicada ao projeto, e é amplamente usado em universidades para ensino de cálculo numérico.

Entre seus recursos computacionais está a facilidade de realizar desde cálculos básicos até simulações complexas, incluindo cálculos matriciais e suportando álgebra linear, estatística, otimização, simulação de sistemas dinâmicos. Também oferece gráficos 2D e 3D para visualização de dados, e possui uma linguagem própria de programação, de alto nível e interpretada, semelhante ao MATLAB. Suas características básicas são as seguintes:

- É multiplataforma: funciona em Windows, Linux e MacOS.
- Apoia a pesquisa científica por ser gratuito e acessível.
- Facilita a criação de modelos matemáticos e análise de dados.
- Pode ser integrado com Fortran, C, C++ e Java.
- Inclui Xcos, ferramenta de simulação de sistemas gráficos equivalente ao Simulink do MATLAB.

SciLab é uma ferramenta pertencente à família dos *softwares* de computação numérica, baseada no MATLAB; entretanto, uma das diferenças fundamentais entre essas duas ferramentas reside no fato de que, enquanto o MATLAB é um *software* comercial pago o SciLab é *open source* e *free*, distribuído sob licença de uso. A **figura 1** mostra o seu ambiente interativo de desenvolvimento, na versão mais atual (2025.1.0); mas o usuário deve entender que o *layout* da ferramenta pode mudar nas versões futuras, além de mudanças profundas em funções, com algumas sendo simplesmente excluídas e/ou substituídas por outras; é o caso, por exemplo, da função **eval()** (para conversão de tipos para *string*), que na versão atual **2025.1.0** foi substituída pela função **evstr()**.

O programa (*script*) “**EstatisticaDeClasses**” apresenta uma solução computacional usando o SciLab, em uma situação em que a **tabela 1** mostra os dados obtidos (em hectares) de um levantamento de terras produtivas em cento e cinco propriedades cadastradas; esses dados, obtidos experimentalmente, são armazenados em um arquivo de texto (**Amostra94.txt**) que é lido. O programa determina o Rol (conjunto de todos os dados da amostra organizados em ordem crescente), determina o número de classes, a amplitude do intervalo de classes e constrói a tabela de distribuição de frequências. A **figura 1a** apresenta o IDE da versão 2025.1.0 e a **figura 1.b** mostra o código (*script*) do programa no editor *SciNotes* da ferramenta. A **figura 2** exhibe a saída do programa: dados originais (lidos do arquivo), dados ordenados, as classes formadas com as amplitudes calculadas e as classes com os respectivos resultados estatísticos: classes, frequências normais, pesos médios, frequências relativas, percentuais e frequências absolutas.

Continua na Parte II

O livro “**SciLab: Uma Abordagem Prática e Didática**”, do autor, pode ser uma grande ajuda para os programadores iniciantes que querem aprender sobre esta ferramenta.

Veja mais detalhes deste livro no *link*:

<https://www.amazon.com.br/SciLab-Uma-Abordagem-Pr%C3%A1tica-Did%C3%A1tica/dp/8539906570>

Amostra.txt - Bloco de notas

Arquivo	Editar	Formatar	Exibir	Ajuda										
72	84	73	71	51	64	73	71	69	74	77	75	82	69	72
66	73	71	92	58	74	82	53	66	76	73	93	80	74	77
76	88	73	62	78	74	70	77	75	74	99	78	62	77	71
53	75	66	83	62	95	68	90	76	73	72	78	69	78	65
65	87	76	72	65	69	81	47	84	43	76	72	78	71	73
64	77	80	68	62	73	70	46	58	71	91	78	67	68	62
90	65	67	57	64	80	66	79	67	69	81	74	67	75	79

Tabela 1 - Conteúdo do arquivo “Amostra.txt”




Figura 1a - O Ambiente de desenvolvimento interativo do Scilab

EstatisticaDeClasses.sce (D:\Cantinho da Programação\Códigos\SciLab\EstatisticaDeClasses.sce) - ...

Arquivo Editar Formatar Opções Janela Executar ?


EstatisticaDeClasses.sce (D:\Cantinho da Programação\Códigos\SciLab\EstatisticaDeClasses.sce) - SciNotes

EstatisticaDeClasses.sce

```
1 //EstatisticaDeClasses.sce
2 //Lê um arquivo texto com dados originais e faz estatísticas de classes
3 //sobre esses dados lidos.
4 //-----
5
6 clc; //limpa a tela
7 clear; //limpa todas as variáveis da memória
8
9 //-----
10 //1. Lê os dados
11 //-----
12 MD = fscanfMat("D:/Livros/Livroll/Amostra.txt"); // lê os números do TXT
13 DB = MD(:); // transforma em vetor coluna
14 n = length(DB); // obtém o tamanho da amostra
15 .....
16 //-----
17 //2. Mostra dados brutos
18 //-----
19 disp("Dados brutos -- desordenados");
20 for i=1:n
21     if modulo(i,15)==0 then
22         printf("%d\n", DB(i))
23     else
24         printf("%d.", DB(i))
25     end
26 end
27 printf("\n-----\n")
28
29 //-----
30 //3. Ordena o vetor de dados (Rol)
31 //-----
32 for i=1:(n-1)
33     for j=(i+1):n
34         if DB(i) > DB(j) then
35             aux = DB(i);
36             DB(i) = DB(j);
37             DB(j) = aux;
38         end
39     end
40 end
```

Linha 85, coluna 18.

Figura 1b - O script do programa editado no SciNotes da ferramenta



The image shows a screenshot of the Scilab 2025.1.0 Console window. The window has a title bar with the Scilab logo and the text "Scilab 2025.1.0 Console". Below the title bar is a menu bar with the following items: "Arquivo", "Editar", "Controle", "Aplicativos", and "?". Below the menu bar is a toolbar with various icons for file operations, editing, and help. The main area of the window displays the output of a program, which is a statistical analysis of a dataset. The output is organized into several sections, separated by dashed lines. The first section is titled "Dados brutos - desordenados" and displays a 4x10 grid of numbers. The second section is titled "Dados ordenados - Rol" and displays a 4x10 grid of numbers. The third section is titled "Classes" and displays a single line of class intervals. The fourth section is a table with 6 columns: "Classe", "Freq.", "Pm", "Fr", "Fp(%)", and "Fa". The table contains 6 rows of data. At the bottom of the window, there is a prompt "--> |" and a horizontal scrollbar.

```
Scilab 2025.1.0 Console
Arquivo  Editar  Controle  Aplicativos  ?
Scilab 2025.1.0 Console
"Dados brutos - desordenados"
72 66 76 53 65 64 90 84 73 88 75 87 77 65 71
92 62 83 72 68 57 64 74 74 95 69 73 80 71 53
77 90 47 46 79 74 76 74 73 43 71 69 75 93 78
78 72 78 74 69 74 77 78 71 68 75
-----
"Dados ordenados - Rol"
43 46 47 53 53 57 62 64 64 65 65 66 68 68 69
69 69 71 71 71 71 72 72 72 73 73 73 74 74 74
74 74 74 75 75 75 76 76 77 77 77 78 78 78 78
79 80 83 84 87 88 90 90 92 93 95
-----
"Classes"
43-50.428571  50.428571-57.857143  57.857143-65.285714  65.285714-72.714286  72.714286-80.142857
-----
"Classe      Freq.      Pm      Fr      Fp(%)  Fa"
43-50.428571   3      46.7      0.05      5.4    3
50.428571-57.857143   3      54.1      0.05      5.4    6
57.857143-65.285714   5      61.6      0.09      8.9   11
65.285714-72.714286  13      69.0      0.23     23.2   24
72.714286-80.142857  23      76.4      0.41     41.1   47
80.142857-87.571429   3      83.9      0.05      5.4   50
87.571429-95       6      91.3      0.11     10.7   56
--> |
<----->
```

Figura 2 - Saída do programa

Código completo do programa -----

```
//EstatisticaDeClasses.sce
//Lê um arquivo texto com dados originais e faz estatísticas de classes sobres
//esses dados lidos do arquivo.
//-----

clc;    //limpa a tela

clear; / Limpa todas as variáveis da memória

// -----
// 1. Lê os dados
// -----
MD = fscanfMat("D:/Livros/Livroll/Amostra.txt"); // lê os números do TXT
DB = MD(:); // transforma em vetor coluna
n = length(DB); // obtém o tamanho da amostra

// -----
// 2. Mostra dados brutos
// -----
disp("Dados brutos - desordenados");
for i=1:n
    if modulo(i,15)==0 then
        printf("%d\n", DB(i))
    else
        printf("%d ", DB(i))
    end
end
printf("\n-----\n")

// -----
// 3. Ordena o vetor de dados (Rol)
// -----
for i=1:(n-1)
    for j=(i+1):n
        if DB(i) > DB(j) then
            aux = DB(i);
            DB(i) = DB(j);
            DB(j) = aux;
        end
    end
end

// -----
// 4. Mostra dados ordenados
// -----
disp("Dados ordenados - Rol");
for i=1:n
    if modulo(i,15)==0 then
        printf("%d\n", DB(i))
    else
        printf("%d ", DB(i))
    end
end
printf("\n-----\n")
```

```

// -----
// 5. Determina parâmetros básicos
// -----
K = round(1 + log(n)/log(2)); // número de classes
A = DB($) - DB(1); // amplitude da amostra
h = A/K; // amplitude de cada classe
C0 = DB(1); // limite inferior mínimo dos dados

// -----
// 6. Cria as classes
// -----
disp("Classes");
//Classe = strings(1,K); // cria vetor de K strings vazias (versões antigas)
Classe = string(zeros(1,K)); // cria vetor de K strings vazias (versão 2025.1.0)
for i=1:K
    if i==1 then
        Cinf = C0;
        Csup = C0 + h;
    else
        Cinf = Csup;
        Csup = Cinf + h;
    end
    Classe(i) = string(Cinf) + "-" + string(Csup);
    printf("%s ", Classe(i))
end
printf("\n-----\n")

// -----
// 7. Determina os limites de cada classe
// -----
Linf = zeros(1,K);
Lsup = zeros(1,K);
for i=1:K
    limites = tokens(Classe(i), "-"); // separa string pelo hífen
    Linf(i) = evstr(limites(1));
    Lsup(i) = evstr(limites(2));
end

// -----
// 8. Frequências
// -----
Freq = zeros(1,K); // absoluta
Freqr = zeros(1,K); // relativa
Freqp = zeros(1,K); // percentual
Pm = zeros(1,K); // ponto médio

for i=1:K
    Soma = 0;
    for j=1:n
        if DB(j)>=Linf(i) & DB(j)< Lsup(i) then
            Soma = Soma + 1;
        end
    end
    //Ajusta última classe
    if i==K & DB(n)>= Lsup(i) then
        Soma = Soma + 1;
    end
    Freq(i) = Soma;

```

```

    Pm(i) = (Linf(i)+Lsup(i))/2;
    Freqr(i) = Freq(i)/n;
    Freqp(i) = Freqr(i)*100;
end

// -----
// 9. Monta a tabela de distribuição
// -----
disp("Classe      Freq.      Pm      Fr      Fp(%)  Fa");
SomaF = 0;
for i=1:K
    SomaF = SomaF + Freq(i);
    printf("%s      %d      %5.1f      %5.2f      %5.1f      %d\n", Classe(i), Freq(i), Pm(i),
    Freqr(i), Freqp(i), SomaF);
end //fim do programa "EstatisticaDeClasses"

```