Produto Vetorial

Mário Leite

...

O Produto Vetorial, ao contrário do produto escalar, produz um vetor: não um escalar. Na Física o produto vetorial (produto externo) é um conceito muito importante; por exemplo, no eletromagnetismo a força exercida sobre uma partícula num campo magnético uniforme é proporcional ao produto vetorial do *vetor* velocidade da partícula pelo *vetor* campo magnético. Outras grandezas vetoriais da Física podem ser citadas: força, campo elétrico velocidade e aceleração; são todas consideradas vetores, que além do *módulo* possuem mais três propriedades: *direção*, *sentido* e *ponto de aplicação*. Entretanto, por fugir ao escopo desta postagem, aqui vamos tratar os vetores de maneira analítica e não como grandezas físicas. Trataremos como elementos matemáticos com valores (coordenadas) nas três dimensões a partir da definição geométrica, tal como foram apresentados na postagem sobre "Produto Escalar". Assim, ainda considerando os vetores **U** e **V** da postagem anterior, o produto vetorial (Pv) pode ser definido analiticamente como o produto dos seus módulos pelo *seno* do ângulo θ entre eles, baseado no esquema da **figura 1**.

$$Pv = \mathbf{U} \times \mathbf{V} = U^*V^*sen(\theta)$$

Onde o símbolo **x** é o operador que indica "produto vetorial" ente os vetores **U** e **V**. E como casos particulares temos os seguintes: o produto vetorial será 0 (zero) quando estiverem sobre uma mesma reta e igual ao produto de seus módulos (U*V) quando são ortogonais entre si.



Figura 1 - Relação entre dois vetores

E como foi mostrado na postagem anterior, o "Produto Escalar" é um escalar (número), mas o "Produto Vetorial" é um vetor (não um número). A propriedade *comutativa* da multiplicação não se aplica a este tipo de produto entre vetores; portanto: Pe = **U** o **V** = **V** o **U**; mas, **Pv** = **U** x **V** ≠ **V** x **U.** A **figura 2** mostra um sistema de eixos ortogonais no espaço, constituído de três eixos direcionais X, Y e Z; também são mostrados três pequenos vetores **i**, **j**, **k**; *vetores unitários* (de módulo 1).



Figura 2 - Vetores unitários no sistema de eixos ortogonais (X,Y,Z)

A **figura 3** mostra o que se convencionou chamar de "*regra da mão direita*" para indicar o sentido do produto vetorial, cujo resultado é um vetor perpendicular ao plano que define os vetores **U** e **V**.



Figura 3 - Regra da mão direita no Produto Vetorial

O esquema da **figura 3** também tem como consequência que o produto vetorial é *anti-comutativo*, tal que: (**U**x**V**) = – (**V**x**U**).

Considere os vetores **U** e **V** de coordenadas espaciais $\mathbf{u_1}$, $\mathbf{u_2}$, $\mathbf{u_3}$ e $\mathbf{v_1}$, $\mathbf{v_2}$, $\mathbf{v_3}$, respectivamente, nos eixos X, Y e Z com os vetores unitários **i**, **j, k** que foram mostrados na **figura 2**,

$$\mathbf{U} = \{u_1\mathbf{i},\ u_2\mathbf{j},\ u_3\mathbf{k}\}$$

$$\mathbf{V} = \{v_1\mathbf{i},\ v_2\mathbf{j},\ v_3\mathbf{k}\}$$

O Produto Vetorial pode ser definido em função de uma matriz 3x3 com a primeira linha formada pelos vetores unitários **i**, **j**, **k** como no esquema abaixo.

$$\boxed{\mathbf{U} \times \mathbf{V} \quad = \quad \begin{array}{ccc} \mathbf{i} & \mathbf{j} & \mathbf{k} \\ u_1 & u_2 & u_3 \\ v_1 & v_2 & v_3 \end{array}}$$

Para calcular o produto vetorial basta aplicar o "Método de Sarrus" para resolver a expressão matricial e obter o determinante da matriz.

$$\begin{array}{ccccc} \mathbf{i} & \mathbf{j} & \mathbf{k} & \mathbf{i} & \mathbf{j} \\ u_1 & u_2 & u_3 & u_1 & u_2 \\ v_1 & v_2 & v_3 & v_1 & v_2 \end{array}$$

O produto vetorial será um vetor **Pv** de três elementos direcionais e dado por:

$$\mathbf{Pv} = [(u_2 * v_3)\mathbf{i} + (u_3 * v_1)\mathbf{j} + (u_1 * v_2)\mathbf{k}] - [(u_2 * v_1)\mathbf{k} + (u_3 * v_2)\mathbf{i} + (u_1 * v_3)\mathbf{j}]$$

$$= (u_2 * v_3 - u_3 * v_2)\mathbf{i} + (u_3 * v_1 - u_1 * v_3)\mathbf{j} + (u_1 * v_2 - u_2 * v_1)\mathbf{k}$$

E como o módulo (norma, intensidade) dos três vetores unitários (**i j k**) é igual a **1**, então o produto vetorial **PV** pode ser dado em função de seus três elementos: Pv = Pv[1], Pv[2], Pv[3] onde estes elementos serão dados, respectivamente, por:

$$Pv[1] = (u_2 * v_3 - u_3 * v_2)$$

$$Pv[2] = (u_3 * v_1 - u_1 * v_3)$$

$$Pv[3] = (u_1 * v_2 - u_2 * v_1)$$

O programa **"ProdutoVetorial"** pode ser uma solução para calcular o produto vetorial entre dois vetores. A codificação deste programa foi feita em Visualg, por ser esta linguagem (aliás, uma ferramenta de ajuda na programação) em que as linhas de instruções podem ser facilmente convertidas para qualquer linguagem. A **figura 4** mostra o resultado do produto vetorial **U**x**V** e a **figura 5** o resultado de **V**x**U**. A **figura 6** mostra um exemplo em que o produto vetorial é nulo; neste caso, os dois vetores são **Colineares** (ângulo entre os vetores: 0º); e a **figura 7** um exemplo em que os dois vetores são **Ortogonais** onde o módulo (norma) do produto vetorial é igual ao produto dos módulos de **U** e **V** e ângulo entre eles de 90º..



Figura 4- Produto Vetorial: UxV



Figura 5 - Produto Vetorial: VxU



Figura 6 - Produto Vetorial: Vetores Colineares



Figura 7 - Produto Vetorial: Vetores Ortoginais

```
Algoritmo "ProdutoVetorial"
//Faz o "Produto Vetorial" de dois vetores no R³.
//Em Visualg
//Autor: Mário Leite
//-------------------------------------------------------------
   Var U, V, Pv: vetor[1..3] de inteiro
       i: inteiro
Inicio
   {Leitura do primeiro vetor}
   Escreval("Leitura dos elementos do primeiro vetor")
   Para i De 1 Ate 3 Faca
      Escreva("Digite o elemento [",i,"] do vetor U: ")
      Leia(U[i])
   FimPara
   Escreval("")
   {Leitura do segundo vetor}
   Escreval("Leitura dos elementos do segundo vetor")
   Para i De 1 Ate 3 Faca
      Escreva("Digite o elemento [",i,"] do vetor V: ")
      Leia(V[i])
   FimPara
   Escreval("")
   Escreval("")
   {Cria o Produto Vetorial UxV}
   Pv[1] <- U[2]*V[3] - U[3]*V[2]
   Pv[2] <- U[3]*V[1] - U[1]*V[3]
   Pv[3] <- U[1]*V[2] - U[2]*V[1]
   {Exibe o Produto Vetorial UxV}
   Escreva("Produto Vetorial UxV = ")
   Escreva("(",Numpcarac(Pv[1]) + "i",", ")
   Escreva(Numpcarac(Pv[2]) + "j",", ")
   Escreva(Numpcarac(Pv[3]) + "k)")
   Escreval("")
 FimAlgoritmo
```