

# Produto Escalar Vetores

Mário Leite

...


O Produto Escalar é definido na Álgebra Linear como uma função binária entre dois vetores fornecendo um número real chamado “escalar”; isto é, para a Matemática o “Produto Escalar” dá como resultado um número real. Já para a Física, embora continue oferecendo um número real como resultado, é aplicado na definição de algumas grandezas físicas como, por exemplo: energia, trabalho, temperatura, massa, distância, etc. Este conceito é muito importante tanto na Física como na Engenharia. Por exemplo, uma carroça puxada por uma força **F** em uma distância **d** realiza um trabalho **T** definido pelo produto do *módulo* dessa força pela distância:  $T = F \cdot d$ . E se o módulo da força for dado em N (*newtons*) e a distância em metros (*m*) o trabalho será medido em J (*joules*) no Sistema Internacional de Medidas físicas, representando uma grandeza física.

**Nota:** Por ser um assunto muito profundo da Física-Matemática e por fugir ao escopo desta postagem, não serão discutidos aqui detalhes teóricos sobre o “Produto Escalar”. O objetivo aqui é mostrar, na prática, como esse tipo de produto pode ser calculado por um algoritmo bem simples e apresentado sob a forma de pseudocódigo, sem precisar de definições matemáticas complexas. Aqui vamos considerar vetores como elementos no espaço  $R^3$  com valores nas três dimensões.

Considere **U** e **V** dois vetores mostrados na **figura 1**; analiticamente, o produto escalar entre eles, denotado por **Pe**, é dado em função de seus módulos e o *coseno* do ângulo **θ** (*teta*) entre eles.

$$Pe = \mathbf{U} \circ \mathbf{V} = U \cdot V \cdot \cos(\theta)$$

Onde o símbolo **o** é o operador que indica “produto escalar”.



**Figura 1 - Relação entre dois vetores**

Podem ser considerados dois casos particulares: se os dois vetores são perpendiculares ao mesmo plano o produto escalar entre eles é 0 (zero); se estiverem sobre uma mesma reta o produto escalar será igual ao produto de seus módulos:  $U \cdot V$  (vamos considerar o módulo do vetor com a letra normal, sem negrito). O “Produto Escalar” é uma operação *comutativa*:  $\mathbf{U} \circ \mathbf{V} = \mathbf{V} \circ \mathbf{U}$ .

Do ponto de vista da Álgebra Computacional é possível calcular o “Produto Escalar” somando os produtos de cada elemento respectivo dos dois vetores. Considere os dois vetores citados, com seus respectivos elementos em cada dimensão:  $\mathbf{U} = \{3, 5, 9\}$  e  $\mathbf{V} = \{2, 4, 6\}$  esquematizados da **tabela 1**.

<b>V/U</b>	<b>3 = 6</b>	<b>5 = 20</b>	<b>9 = 54</b>	<b>Pe = 80</b>
<b>2</b>				
<b>4</b>				
<b>6</b>				

**Tabela 1 - Esquema do “Produto Escalar” entre os vetores U e V**

Considerando o esquema apresentado na **tabela 1** o “Produto Escalar” entre os vetores **U** e **V**, ambos com **m** elementos (do mesmo tamanho), basta somar os produtos dos seus respectivos elementos.

$$Pe = U[1]*V[1] + U[2]*V[2] + U[3]*V[3] + \dots + U[m]*V[m]$$

Resumindo: é possível escrever a expressão acima de forma compacta para o “Produto Escalar” entre dois vetores **U** e **V** (em **negrito**) de **m** elementos em função da expressão da **figura 2**.


$$\mathbf{U} \circ \mathbf{V} = \sum_{i=1}^{i=m} U[i] * V[i]$$

**Figura 2 - Expressão do “Produto Escalar” entre dois vetores**

Agora, baseando no esquema da expressão da **figura 2** é possível escrever um programa para calcular o “Produto Escalar” entre dois vetores. O programa “**ProdutoEscalar**” (em pseudocódigo que pode ser convertido em qualquer linguagem de programação) mostra como fazer isto para dois vetores de três elementos cada um, tal como apresentados na **tabela 1**.

```
Programa "ProdutoEscalar"
//Calcula e exibe o "Produto Escalar" entre dois vetores
//Em Pseudocódigo
//Autor: Mário Leite
//-----
Const RD2GR=57.29577951308232 //converte radianos para graus
Declare U, V: arrays[1..3] de inteiro
ModU, ModV, AngUV: real
i, Pe: inteiro
Início
Para i De 1 Até 3 Faça //leitura do vetor U
    Escreva("Digite o elemento [",i,"] do vetor U: ")
    Leia(U[i])
FimPara
EscrevaLn("") //salta linha
Para i De 1 Até 3 Faça //leitura do vetor V
    Escreva("Digite o elemento [",i,"] do vetor V: ")
    Leia(V[i])
FimPara
EscrevaLn("")
EscrevaLn("")
{Calcula e exibe o produto escalar UoV}
Pe ← 0
Para i De 1 Até 3 Faça
    Pe ← Pe + U[i]*V[i]
FimPara
EscrevaLn("Produto Escalar UoV: ", Pe)
{Calcula e exibe o ângulo entre os vetores}
EscrevaLn("")
ModU ← RaizQ(U[1]^2 + U[2]^2 + U[3]^2)
ModV ← RaizQ(V[1]^2 + V[2]^2 + V[3]^2)
AngUV ← Pe/(ModU*ModV)
AngUV ← (ArcCos(AngUV))*RD2GR //ângulo em graus
EscrevaLn("Ângulo entre os dois vetores: ", AngUV, "graus")
EscrevaLn("")
FimPrograma
```

A saída do programa “**ProdutoEscalar**”, codificado em Visualg, é mostrada na **figura 3** que, além de calcular o “Produto Escalar” também mostra o valor do ângulo entre eles,



```
C:\ Console simulando o modo texto do MS-DOS

Digite o elemento [ 1] do vetor U: 3
Digite o elemento [ 2] do vetor U: 5
Digite o elemento [ 3] do vetor U: 9

Digite o elemento [ 1] do vetor V: 2
Digite o elemento [ 2] do vetor V: 4
Digite o elemento [ 3] do vetor V: 6

Produto Escalar UoV: 80

Ângulo entre os dois vetores: 4.52023 graus

>>> Fim da execução do programa !
```

**Figura 3 - Saída do programa “Produto Escalar” em Visualg**